
�Introduzione.





Questo documento costituisce un veloce riferimento alla sintassi del linguaggio Prolog ed a molte altre informazione che spesso sono necessarie quando si programma in questo linguaggio. Sono anche forniti alcuni esempi di implementazione di algoritmi molto diffusi e utili. Questo documento non è assolutamente completo ed esauriente su molti aspetti del linguaggio, ma è una specie di “appunto veloce e generico” che ho compilato in occasione del compito di Intelligenza Artificale... :-).

Alcuni dei codici contenuti sono stati reperiti da diverse fonti ed appartengono ai rispettivi proprietari, altri li ho scritti io.

Tra le fonti principali:

- Guida dell’ Amzi! Logic Explorer

- Prolog tutorial di James Power

- Prolog tutorial di J.R.Fisher

- LOGIC, PROGRAMMING AND PROLOG (2ED) di Ulf Nilsson and Jan Maluszynski (disponibile on line)



Spero che possa esservi utile quanto lo è stato a me.





Palermo 26/11/2001

Antonio Di Stefano

ragazzoionico@tin.it

P.S. Mi scuso per il mix di italiano ed inglese :-))

�Complex goals



X , Y

X , Y succeeds if both X and Y both can be proved; else it fails.



X ; Y

X ; Y succeeds if X can be proved or Y can be proved; else it fails.



Goal1 -> Goal2 ; Goal3

Goal1 -> Goal2 ; Goal3 is an if-then-else construct. If Goal1 can be proved then Prolog tries to prove Goal2. Otherwise if Goal1 fails Prolog tries to prove Goal3. Goal1 is not backtrackable into once it has been proved.



This predicate can also be used in the form Goal1 -> Goal2, which only succeeds once, if at all, and if so executes Goal2.



call(Goal)

call succeeds if Goal can be proved. Goal must be instantiated to a term which could be a valid goal in a clause body. Then call succeeds if and only if that goal can be proved. Note that Goal may be provable using compiled code or dynamic clausesthe call predicate handles both with no need for declarations.



not(Goal)

not succeeds if and only if Goal cannot be proved. Gola is subject to the same restrictions on Goal as in call above.



Note that not(not(Goal)) has the interesting, and sometimes useful, effect of succeeding if Goal can be proved, but not unifying any of its variables and failing on backtracking.



\+ Goal

\+ X is a synonym for not X.



findall(Instance, Goal, List)

findall succeeds if List can be unified with the list of all instances of Instance making Goal provable.



For example, findall(X, a(X), [1, 2, 3]) is true if the database contains precisely the following clauses for a:



a(1). 

a(2). 

a(3).



If Goal cannot be proved for any values of Instance, then List is unified with the empty list [].



findall is generally used to generate lists from database entries, so for example it might be used as follows.



?- findall(X, a(X), L).

L = [1, 2, 3]



bagof(Instance, Goal, List)

bagof is like findall above except in the way it deals with variables occurring in Goal which are not in Instance. These are known as free variables. In this case, bagof is backtrackable into and produces one list List for each possible binding of the free variables.



It is possible to convert an otherwise free variable to a non-free variable by using the ^ symbol as follows:



bagof(W, a(W, X, Y, Z), L).		

% Here X, Y and Z are free



bagof(W, X ^ Y ^ a(W, X, Y, Z), L)	

% Here only Z is free



So, for example, consider the following database:



likes(fred, beer).

likes(tom, wine).

likes(jane, beer).

likes(jane, coke).

?- bagof(P, likes(X, P), L).   % X is free so we will backtrack 

P = H47

X = fred

L = [beer] ;

P = H47

X = tom

L = [wine] ;

P = H47

X = jane

L = [beer, coke] ;

no

?- bagof(P, X ^ likes(X, P), L).

P = H47

X = H48

L = [beer, wine, beer, coke].



get_preds(PredList)

get_preds/1 is called with an unbound variable, PredList, and returns a list with the names and arities of all the predicates in the dynamic database. The predicates are returned in the form: functor/arity.  For example, if DUCKY.PRO were loaded in the listener, then:



?- get_preds(X).

X = [ main / 0, goto / 1, ...



once(Goal)

The predicate once behaves like call except it is only executed once.  This is useful for isolating code that you don't intend to be backtracked into, and makes for more readable code than inserting lots of cuts (!).



Note that like all such predicates, more than one goal can be given as an argument by enclosing them in parenthesis:



main :-



    first_backtracking_goal,

    once(

    (

      first_once_goal,

      second_once_goal,

      third_once_goal

    )),

    next_backtracking_goal,

    ...



setof(Instance, Goal, List)

setof is like bagof/3 above except the list List is sorted according to the standard order and any duplicates are removed.





Flow control predicates



The top-down, left-to-right search coupled with backtracking will try to ferret out a solution to a problem. Sometimes it can be too industrious. What we need is a way of saying “if you have got this far then there is no backtracking !” 



!Cut

There is a Prolog predicate which does just this. It is called cut, and is denoted with a “!“.



! is always true, so if a clause containing a cut is read as a statement of truth, it behaves as if there were no cut there. But cut affects the way backtracking is performed as follows:

Once a cut is executed, the choice of the clause which contains it is frozen as a proof step. Also any choices made during the proof of the goals between the head of the clause and the cut are frozen. Thus cut acts like a fence. When backtracking passes over the cut (heading left in a clause), then proof reconsideration continues not with the goal to the left of the !, but the goal to the left of the goal which chose the clause containing the cut.



repeat

repeat is always provable, and can be backtracked into an arbitrary number of times. It behaves as though it had been defined by:



repeat.

repeat :- repeat.



fail

fail always fails.



true

true always succeeds.



for(Index, Start, End, Increment)

for provides a shorthand for implementing repeat/fail loops that execute a prespecified number of times.  Bottom, Top and Increment must be bound to integers with Bottom being less than or equal to Top. When first proved Index is unified with Bottom and checked to see whether it is less than or equal to Top. If so, for succeeds otherwise it fails. 



On backtracking Increment is added to the current value of Index and Index is bound to this new value. Again a range check is performed.



?- for(X, 1, 5, 1), write(X), fail.



12345no



catch(Goal, Catcher, Recover) and throw(Term)

Two predicates, catch/3 and throw/1, are a general purpose mechanism for handling user-defined exceptions in Prolog code.   These predicates are the ISO standard predicates for exception handling, and provide an alternative to the current tag/1 and cut_tag/1 predicates.



The two key predicates' arguments are:

catch(Goal, Catcher, Recover)



Goal	is a normal Prolog goal to be proved

Catcher	is a term used as a pattern to be checked against possible thrown exceptions

Recover	is a goal to be proved if a thrown exception is caught.



throw(Term)



Term	a term that is used in a search for a matching catch term



catch(Goal,Catcher,Recover) is fully equivalent to ‘

call(Goal)’ for the normal case where no exception is thrown during the execution of goal Goal.  If a ‘

throw(Catcher)’ is encountered, then the catch/3 Goal is replaced with the goal ‘call(Recover)’ instead.



The catch/throw pair allow the flow-of-control to skip over all of the intermediate predicates, which is especially useful for dealing with error exceptions.

Throw will only succeed if the term thrown matches an active catch term. After the goal of the catch is satisfied the catch term can no longer be thrown to. An attempt to throw an uncaught term results in a system error.

The following is the example from CATCH.PRO in the PROSAMP directory:



main :-



	/* Catch and process all throws not handled by subsequent catches, including throw(quit) used to end the program. */



	   catch(doit, X, error_handler(X)).



error_handler(quit) :-

	  write($Quitting\n$).

error_handler(X) :-

	  write($Unknown Error$ : X), nl.



doit :-

	  repeat,

	  write($Enter one or done or something else: $),

	  read_string(S),

	  string_atom(S, A),

	  catch(do(A), badcommand(X),

       (write($Bad Command$:X),nl)),

	  fail.



do(one) :-

	  catch(do_one, except(X), except(X)), !.

do(done) :-

	  throw(quit).

do(X) :-

	  throw(badcommand(X)).



except(notinteger:X) :-

	  write(X), write($ not an integer\n$).

except(toosmall:X) :-

	  write(X), write($ is too small\n$).

except(toobig:X) :-

	  write(X), write($ is too big\n$).



do_one :-

	  repeat,

	  write($Enter a number between 10 and 20,\n$),

	  write($'q' to quit,



      or anything else to see an   error:\n$),

	  read_string(S),

	  string_term(S,T),

	  check(T),

	  fail.



check(q) :- throw(quit).

check(X) :-

	  not(integer(X)),

	  throw(except(notinteger:X)).

  check(X) :-

	  X > 10,

	  X < 20,

	  !, write($Very good\n$).

check(X) :-

	  X =< 10,

	  throw(except(toosmall:X)).

check(X) :-

	  X >= 20,

	  throw(except(toobig:X)).

check(X) :-

	  throw(badcheck(X)).



Note that multiple goals can be provided to catch/3

, such as:



main :-



     catch(

          (goal1,

           goal2,

           goal3),

     error(X,Y),

     process_error(X,Y)).



tag(Term) and cut_tag(Term)



®	Tags are an advanced feature of Amzi! Prolog. See also the ISO standard predicates catch/3 and throw/1.



! is very useful to control backtracking behavior in Prolog programs but its effects are confined only to the code of the predicate in which the clause containing the cut is a part. There are occasions in which we would like to modify backtracking behavior beyond the current predicate— an example of which is in handling errors.



To facilitate a form of more general backtracking control Amzi! Prolog has introduced the concept of the tag.

A tag is a marked instance of a choice point in the execution environment of a Prolog program. The choice point is marked by associating it with some term using the predicate tag/1. The predicate cut_tag/1 allows the removal of all choice points back to and including the choice point marked by a given term.

tag associates the most current choice point with Term. This association is only removed when the tagged choice point is removed by backtracking.



cut_tag scans back through the choice points until it comes across the first choice point with a tag which can be unified with Term. All choice points more recent than the tagged choice point are removed.

As an example, consider the use of tags to “fail out of” the infinite loop of a repeat (repeat is a special built in predicate which can always be reproved an unlimited number of times) so 



repeat, do, fail 



acts like an infinite loop.



loop :-



tag(loop_marker), 

	repeat, 

	doit, 

	fail.

doit :- want_to_stop, cut_tag(loop_marker), !.

doit :- do_something.



When the clause for loop is entered it will tag the most recent choice point with label “loop_marker.” 



Then the repeat will be proved and doit will be tried. As long as want_to_stop is not provable, do_something will be proved and then fail will be tried. fail always fails, so backtracking takes us eventually back to repeat and doit is proved again. 

Once we can prove want_to_stop, cut_tag

(loop_marker) is proved. This looks back along the proof and finds that the choice point before repeat was marked with loop marker. So every choice point since then (including the choice point for repeat) is removed. Finally fail is proved one more time. But now the repeat has been effectively removed and so backtracking will go all the way back to some point before loop was entered.



halt

When proved, halt returns to the operating system. halt will flush any files to disk and close them so no data will be lost.





Atom classification



The following predicates may be used to classify terms in Prolog:



atom(X)

atom succeeds if X is currently instantiated to an atom; else it fails.



atomic(X)

atomic succeeds if X is currently instantiated to an atom or an integer; else it fails.



float(X)

float succeeds if X is currently instantiated to a float. Can also be used in arithmetic expressions to convert a value to a float.



integer(X)

integer succeeds if X is currently instantiated to an integer. Can also be used in arithmetic expressions to convert a value to an integer.



list(X)

list succeeds if X is currently instantiated to a list; else it fails. If X is the empty list, [ ], list fails.



long(X)

succeeds if X is a long integer



nonvar(X)

nonvar succeeds if X is not an unbound variable; else it fails.



number(X)

number succeeds if X is currently instantiated to an integer or float.



short(X)

succeeds if X is a short integer



string(X)

string succeeds if X is currently instantiated to a string; else it fails.



structure(X)

structure succeeds if X is currently instantiated to a structure; else it fails.



var(X)

var succeeds if X is an unbound variable; else it fails.





Comparison of terms



Two terms may be compared via the standard ordering.



X @< Y

X @< Y succeeds if X is less than Y in the standard order



X @> Y

X @> Y succeeds if X is greater than Y in the standard order



X @=< Y

X @=< Y succeeds if X is less or equal to Y in the standard order



X @>= Y

X @>= Y succeeds if X is greater or equal to Y in the standard order



compare(Result, Term1, Term2)



compare compares terms Term1 and Term2 using the standard ordering. Unifies Result with ==, < or > depending on whether the terms are equal, or Term1 @< Term2 or Term1 @> Term2.





Equality terms



Equality is determined with the following predicates:



X = Y

X = Y succeeds if X can be unified with Y.



X \= Y

X \= Y succeeds if X cannot be unified with Y.



X == Y

X == Y succeeds if X and Y are identical, i.e., they unify with no variable bindings occurring.



X \== Y

X \== Y succeeds if X and Y are not identical.















Dissecting terms



These predicates are useful for breaking apart and constructing generic terms when you don't know the specifics of their structure. A good example of this is a pretty printer.



arg(N, Term, Argument)

N must be instantiated to a positive integer less than or equal to the number of arguments in the compound term Term. Argument is then instantiated to the Nth argument of Term. If all these conditions are not met, the goal fails. For example:



?- arg(1, connect(yard, pen), X).



X = yard



functor(Term, Functor, N)

There are two cases to consider:



·	Term is bound to a structure or an atom. In this case Functor must unify with the principal functor (name) of the structure and N with its arity (or 0, if Term is an atom).



·	Term is an unbound variable. In this case Functor must be bound to an atom and N to a non-negative integer. Term is then unified with the most general term whose principal functor is Functor and whose arity is N (if N is greater than 0).



For example:



?- functor(T, connect, 2), arg(1, T, yard), arg(2, T, pen).



T = connect(yard, pen)



?- functor(connect(yard, pen), F, A).

F = connect

A = 2



gensym(Root, Sym)

gensym/2 is used to create atom names on the fly. The names are formed by adding successive integers to the root. gensym is designed to create new symbols each time it is called; it fails on backtracking. For example



?- gensym(foo, X).



X = foo1

?- gensym(foo, X).

X = foo2 

?- gensym(foo, X).

X = foo3



numbervars(Term, Start, End)

Start must be instantiated to a non-negative integer. Any non-anonymous variables (i.e. those appearing only once) occurring in Term are bound to an atom of the form _I. End is instantiated to Start plus the number of distinct variables in Term.



I is determined by Start and the index of the variable. 

For example:



?- numbervars(b(X, Y), 0, E).



X = _0

Y = _1

E = 2 



Term =.. List (univ)

The operator “=..” is called “univ.” “Term =.. List” converts between compound terms Term and the list List. If Term is instantiated to a compound term, then List is unified with the list whose first element is the principal functor of Term, and whose successive elements are the arguments of Term. An atom is treated as a compound term of arity 0.



If Term is a variable, List must be instantiated to a list of definite length whose first element is an atom. Then Term is unified with the structure whose principal functor is the head of List and whose arguments are the elements in the tail of List.

univ fails if the arguments do not make legal structures.



?- T =.. [as_easy_as, 1, 2, 3].



T = as_easy_as(1,2,3) 



?- as_easy_as(1,2,3) =.. L.

L = [as_easy_as,1,2,3]  



varlist(List)

varlist/1 returns a list of the variable names passed to the current term. Each entry in the list is a list itself. For example:



?- foo(Var1, Var2) :- varlist(X), write(X).



Term asserted

?- foo(A, BB).

[[65], [66, 66]]

A = H30

BB = H31



varsof(Term, List)

varsof/2 succeeds if List can be unified with a list of all uninstantiated (and non-anonymous) variables in Term. Each variable occurs only once in List even though it may have more than one occurrence in Term.



For example:



?- varsof( a(X, Y, X), L).



X = H0

Y = H1

L = [H0, H1] 





Arithmetic



X is Y

Succeeds if X can be unified with the value of Y evaluated as an arithmetic expression.



X >= Y

Greater or equal.



X =< Y

Less than or equal.



X > Y

Greater than.



X < Y

Less than.



X =:= Y

X =:= Y succeeds if X evaluated is arithmetically equal to Y evaluated; else it fails.



X =\= Y

X =\= Y succeeds if X evaluated is not arithmetically equal to Y; else it fails.



Note that in evaluating arithmetic comparisons, X and Y are first evaluated and the appropriate test then succeeds or fails according to the inequalities’ being true or false.

It is a bad Prolog practice to test for numerical (in)equality by using =, \=, == or \==.



X + Y

Sum of values of X and Y.







X - Y

Value of X minus value of Y.



X * Y

Value of X multiplied by value of Y.



X / Y

Value of X divided by value of Y. Always returns a floating point value.



X // Y

Integer division of X by Y—truncates result to the nearest integer.



X mod Y

The remainder after dividing the value of X by the value of Y.



For the following bitwise operators the arguments must be 16 bit integers.



X /\ Y

Bitwise “and” of value of X and value of Y.



X \/ Y

Bitwise “or” of value of X and value of Y.



X << Y

Evaluates to X bit-shifted left by Y places. Note this is an arithmetic shift (does not include the sign bit). So 0 is 1 << 16.



X >> Y

Evaluates to X bit-shifted right by Y places. Again watch the sign bit e.g. -4 >> 1 =:= -2 .



X ** Y

Evaluates to X raised to the Y power.



\ X

Evaluates to the bitwise complement of X (i.e., all those bits which were 1 become 0 and vice versa).



- X

Evaluates to the negative of X evaluated.



acos(X)

acos evaluates to the angle (in radians) whose arccosine is X.



asin(X)

asin evaluates to the angle (in radians) whose arcsine is X.



atan(X)

atan evaluates to the angle (in radians) whose arctangent is X.



cos(X)

cos evaluates to the cosine of X (in radians)



exp(X)

exp evaluates to e raised to the power of X evaluated.



float(X)

float converts X to a floating point number (perhaps with loss of some precision).



integer(X)

integer converts X to an integer (truncating any fractional part).



ln(X)

ln evaluates to the natural log (loge()) of X evaluated.



sin(X)

sin evaluates to the sine of X (in radians).



sqrt(X)

sqrt evaluates to the square root of X.



tan(X)

tan evaluates to the tangent of X (in radians).



Built-in Atoms

There are a number of built-in atoms, which have predetermined values that can be used in arithmetic expression.



cputime 	A floating point number with the number of CPU seconds expired. It is useful for timing functions, for example:

	?- T1 is cputime, dothing, T is T1 - cputime, write(time:T).

e 	The value “e” (2.718282..)

pi 	The value “pi” (3.14159 ..)

random 	A random floating point number between 0 and 1.





Modifying the database



A number of predicates are provided for adding and removing static and dynamic clauses.



abolish(Name/Arity), abolish(Name, Arity)

abolish removes all clauses (dynamic or compiled) defining the predicate whose name is Name and whose arity is Arity using this predicate. For example: 



?- abolish(expenses/3).



asserta(Term)

asserta adds clauses. Term must be sufficiently instantiated to be a term which represents a clause. The clause is added at the front of the database. This predicate fails if the term being asserted is a predicate defined internally by the Prolog system. Such predicates may not be redefined. For example: 



?- asserta(happy).



assertz(Term)

assertz is similar to asserta except the clause is added at the end of the database. Again system predicates may not be redefined. For example: 



?- assertz(customer('John Henry', 'Oshkosh', 'WI')).



assert(Term)

assert is a synonym for assertz.



clause(Head, Body)

clause succeeds if Head can be unified with the head of a dynamic clause in the database, and Body with its body. If clause is backtracked into, it will keep on succeeding as long as there is an untried clause whose head matches Head, in which case it will unify Body with the new body. Eventually the backtracking will fail.



Head must be sufficiently instantiatedthe name and arity of the structure must be apparent in Head. For example:



?- clause(fox, Body).



Body = (loc(ducks, yard) , loc(you, house) , write($ The fox has taken a duck. $) , nl) ;

Body = true



retract(Term)

retract removes clauses. Term must be sufficiently instantiated to represent a clause. In particular the subterm representing the head of the clause must have a well-defined principal functor and number of arguments. The first clause in the database which matches Term is removed from the database. For example: 



?- retract(happy).



retractall(Term)

retractall is similar to retract except all clauses which match Term are removed, and it always succeeds. For example: 



?- retractall(customer(_, _, 'WI')).





Files



This section describes the predicates for working directly with files. Like the streams described above, ID in the predicates below can either be the file handle obtained from fopen, or the file name passed to fopen.



Binary files have an associated file pointer. This is a pointer into the file which indicates where the next byte will be written to or read from. It is normally adjusted after each read or write so that bytes are read or written consecutively.



fopen(Handle, File, Mode)

fopen opens the file with filename File in a specific Mode and returns its Handle. 



File can be an atom, a string, or a character list. Mode should be one of r (read), w (write), a (append), rb (read binary), wb (write binary) or ab (append binary). 

If the binary option is not used then the file system may do some strange things to carriage-return / line-feed pairs of characters. As long as ASCII text is being read (e.g. in a PRO file) then this manipulation is beneficial.

If File cannot be opened then an error is raised (or failure occurs, depending on the fileerrors mode).



fclose(ID)

fclose closes the file corresponding to ID. It raises an error if the file could not be closed. If the file was opened for writing it is flushed before being closed. 



fflush(ID)

fflush flushes pending writes to disk for the file identified by ID, if it has been opened for writing. 



fread(ID, Value, Type)

fread reads data from a file. It is the same as fwrite except Value is unified with the value read from the disk file. If end-of-file is met then fread fails, rather than returning an error condition. 



fseek(ID, Offset, Method, NewOff)

fseek repositions the pointer of the file associated with ID. Offset is a number representing the number of byte positions to be added to the starting point, which is determined by the value of Method as follows: 



0	offset from 0

1 	offset from the current file position

2 	offset from the end of file



NewOff is bound to the new position, and the file is repositioned to point to it. For example, to get the current position of a file:



?- fseek(H, 0, 1, X).



X = 42



fwrite(ID, Value, Type)

fwrite is for random file I/O. Value must be instantiated to a value which is written to the file ID at the current file position. Either one byte is written (Type = 0) or one 16 bit word (Type = 1) or one 32 bit floating point number (Type = 2) or one 32-bit integer (Type = 3).



file_exists(File), file_exists(File, Type)

file_exists/1 succeeds if the File exists. File can be an atom, a string, or a character list specifying the path name of a file. 



file_exists/2 is similar to file_exists/1 above. It succeeds if File exists, unifying Type with 1 if File is a Prolog Object file or with 0 if Type is an ASCII file. Note that file_exists does not examine the whole file before deciding on the type of the file, and in fact will classify any file into one of two types. However if the file is either source or compiled code then Type will accurately indicate which it is.





Keyboard



These predicates read or simulate reading the keyboard.



get1(Key)

get1/1 gets a single character response key for environments that don't support keyb/1. Unlike keyb/1 and friends, get1/1 does not respond until after the [Enter] key is struck. See also respkey/1.



keyb(Ascii)

keyb/1 returns the ASCII character code of the next key struck without echoing to the screen.  keyb/1 is not available in all environments.  It is implemented in DOS, and is available for Windows programs running in the IDE environment, but not for Windows programs running in other environments.



respkey(Char)

respkey/1 gets a single character response key by calling keyb/1 predicate if it exists, and get1/1 if not.  Here is its definition:

respkey(A) :- keyb(A), put(A), nl, !.

respkey(A) :- get1(A).







�Implementazioni varie...





Gestione liste





car([X|Y],X).

cdr([X|Y],Y).

cons(X,R,[X|R]).





% Toglie arg1 dalla lista arg2, arg3 è il resto della lista 

% takeout(3, [2,3,4,5],X).

% X = [2,4,5]

 

takeout(X,[X|R],R).

takeout(X,[F|R],[F|S]) :- takeout(X,R,S).





% Concatena le liste arg1 e arg2, risultato in arg3

% append([1,2],[3,4],X).

% X=[1,2,3,4].



append([], L, L).

append([X | L1], L2, [X | L3]) :- append(L1, L2, L3).



last(X, [X]).

last(X, [_,Y]) :- last(X, Y).



% member(E1, List) :- append(_, [E1 | _], List).





% ribalta la lista arg1 e la restituisce in arg2

% rev([1,2,3],Y).

% Y=[3,2,1]



rev([], []).

rev([H | T], L) :- rev(T, Z), append(Z, [H], L).





% cancella gli elementi arg1 dalla lista arg2 e ritorna la lista in arg3

% delete(b,[a,b,c,b,e],X).

% X=[a,c,e]



delete(_, [], []).

delete(X, [X | L], M) :- !, delete(X, L, M).

delete(X, [Y | L1], [Y | L2]) :- delete(X, L1, L2).





% Sostituisce gli elementi arg1 con arg3 nella lista arg2 ritornata in arg4

% subst(2,[1,2,3,2,4],0,X).

% X=[1,0,3,0,4]



subst(_, [], _, []).

subst(X, [X | L], A, [A | M]) :- !, subst(X, L, A, M).

subst(X, [Y | L], A, [Y | M]) :- subst(X, L, A, M).





% E' verificato se arg1 è un prefisso di arg2

% prefix([1,2],[1,2,3,4]).

& yes



prefix([], _).

prefix([X | L], [X | M]) :- prefix(L, M).





% E' verificato se la lista arg1 è un pezzo di lista arg2

% sublist([2,3],[1,2,3,4]).

% yes



sublist([X | L], [X | M]) :- prefix(L, M), !.

sublist(L, [_ | M]) :- sublist(L, M).





% E' verificato se l'elemento arg1 è membro della lista arg2

% member(2,[1,2,3]).

% yes



member(X, [X|_]).

member(X, [_|Y]) :- member(X, Y).





% arg1 è contenuto in arg2?

% subset([2,3], [1,2,3,4]).

% yes



subset([A|X],Y) :- member(A, Y), subset(X, Y).

subset([], Y).





% intersezione vuota tra arg1 e arg2?

% disjoint([1,2], [3,4,5]).

% yes



disjoint(X, Y) :- not((member(Z, X), member(Z,Y))).





% intersezione tra arg1 e arg2 -> arg3

% intersection([1,2,3],[2,3],X).

% X=[2]



intersection([], X, []).

intersection([X|R], Y, [X|Z]) :-member(X, Y), !, intersection(R, Y, Z).

intersection([X|R], Y, Z) :- intersection(R, Y, Z).





% unione tra arg1 e arg2 -> arg3

% union([1,2],[2,3,4],X).

% X=[1,2,3,4]



union([], X, X).

union([X|R], Y, Z) :- member(X, Y), !, union(R, Y, Z).

union([X|R], Y, [X|Z]) :- union(R, Y, Z).



% aggiungere nth, push, pop e altre funzioni utili







�Ricerca





% Rappresentazione dello stato:

% state(on(a, on(b, on(c, table))), table, table).





% Azioni possibili (operatori)

move(state(on(X, NewX), OldY, Z), state(NewX, on(X, OldY), Z)).

move(state(on(X, NewX), Y, OldZ), state(NewX, Y, on(X, OldZ))).

move(state(X, on(Y, NewY), OldZ), state(X, NewY, on(Y, OldZ))).

move(state(OldX, on(Y, NewY), Z), state(on(Y, OldX), NewY, Z)).

move(state(X, OldY, on(Z, NewZ)), state(X, on(Z, OldY), NewZ)).

move(state(OldX, Y, on(Z, NewZ)), state(on(Z, OldX), Y, NewZ)).





% Predicato di goal (e stato di goal)

% Può essere più di uno!



isgoal(state(on(c, table),on(b, on(a, table)),table)).





% ******

% * Ricerca in ampiezza, con controllo stati visitati

% ******



% --> Invocazione della ricerca

% brefs(start,isgoal,X).

% X=...



brefs(Start, GoalPred, Sol) :- 

  write('apsetta...'), nl,

  bfs([[Start]|Qtail], Qtail, [], GoalPred, Sol).





% --> Ricerca vera e propria



%if the queue is empty, fail 

bfs(OPEN,Qtail, _, _, _) :- 

  OPEN==Qtail, !, 

  fail.

 

%otherwise, as in the previous implementation: 

bfs([[Node|Path]|_], _, _, GoalPred, [Node|Path]) :- 

  Goal =.. [GoalPred, Node], 

  Goal.

 

bfs([[Node|Path]|MoreOPEN], Qtail, CLOSED, GoalPred, Sol) :- 

	findall(

	  [Next,Node|Path], 

	  (

	    move(Node, Next),  % *** predicato operatore ***

	    %not(member([Next|_], [[Node|Path]|MoreOPEN])), 

	    not(member(Next, CLOSED))

	  ), 

	  NewPaths

	  ),

  

	%and here is where the difference list pays off:  

	append(NewPaths, NewQtail, Qtail), 

	bfs(MoreOPEN, NewQtail, [Node|CLOSED], GoalPred, Sol).





% ********

% * Depth-first search, without side effects 

% ********



dpts(Start, GoalPred, Sol) :- 

	dfs([[Start]], [], GoalPred, Sol).



%  The first argument to progfont dfs is the OPEN stack, 

%  the second argument is the CLOSED list 

dfs([[Node|Path]|_], _, GoalPred, [Node|Path]) :- 

	Goal =.. [GoalPred,Node], 

	Goal.

 

dfs([[Node | Path] | MoreOPEN], CLOSED, GoalPred, Sol) :- 

	% find the  new neighbors of the first OPEN node  

	%and add the current path to each of them: 

	findall(

		[Next,Node|Path], 

		(

		  move(Node, Next),

		  %not(member([Next|_], [[Node|Path]|MoreOPEN])), 

		  not(member(Next, CLOSED))

		), 

		NewPaths

		), 

	% place the new paths on top of the stack: 

	append(NewPaths, MoreOPEN, NewOPEN), 

	dfs(NewOPEN, [Node|CLOSED], GoalPred, Sol).





% ******

% * Graph traversal search

% ******



% Invocazione

path(X) :- path(state(on(c, on(b, on(a, table))), table, table),

                state(table, table, on(c, on(a, on(b, table)))), X).

path(X, Y, Path) :- path(X, Y, [X], Path).



path(X, X, Visited, Visited).



% Ricerca vera e propria

path(X, Z, Visited, Path) :-

  move(X, Y),

  not(member(Y, Visited)),

  path(Y, Z, [Y|Visited], Path).







% *** funzioni ausiliarie ***



append([], L, L).

append([X | L1], L2, [X | L3]) :- append(L1, L2, L3).



member(X,[X|Y]).

member(X,[Y|Z]) :- member(X,Z).







% rappresentazione alternativa dello stato



% altro mondo dei blocchi



sposta([[X|L],C,R], [L,[X|C],R]) :- X\=[]. % sposta da sx a cx

sposta([[X|L],C,R], [L,C,[X|R]]) :- X\=[]. % sposta da sx a dx

sposta([L,[X|C],R], [[X|L],C,R]) :- X\=[]. % sposta da cx a sx

sposta([L,[X|C],R], [L,C,[X|R]]) :- X\=[]. % sposta da cx a dx

sposta([L,C,[X|R]], [[X|L],C,R]) :- X\=[]. % sposta da dx a sx

sposta([L,C,[X|R]], [L,[X|C],R]) :- X\=[]. % sposta da dx a cx



% predicato di goal

isgoal([[],[],[a,b,c]]).



% invocazione della ricerca

% brefs([[a,b,c],[],[]],isgoal,X).

% X=...





�Problemi vari...





% Problema dei colori



% Elenco colori

color(red).

color(blue).

color(green).

color(yellow).



% Descrizione delle adiacenze

map([[1,2],[1,3],[1,4],[1,5],[2,3],[2,4],[3,4],[4,5]]).





% Funzione di ricerca (in profondità)  

% N è il numero delle aree



trova(Coda, 0, Coda).



trova(Coda, N, Soluz) :-

  N>0,

  nuovo(Nodo, N),

  not(confl(Nodo, Coda)),

  M is N-1,

  trova([Nodo|Coda], M, Soluz).



% Funzioni ausiliarie



nuovo([N,Colr], N) :- color(Colr).



vicini(X, Y) :-

  map(M),

  member([X,Y], M);

  map(M),

  member([Y,X], M).



confl([RegX|ColX],[[RegY|ColY]|Tail]) :-

  vicini(RegX, RegY),

  ColX=ColY;

  confl([RegX|ColX],Tail).



member(X,[X|Y]).

member(X,[Y|Z]) :- member(X,Z).





% Torre di Hanoi

% Uso: move(3,left,right,center). 



move(1,X,Y,_) :-  

    write('Move top disk from '), 

    write(X), 

    write(' to '), 

    write(Y), 

    nl. 

move(N,X,Y,Z) :- 

    N>1, 

    M is N-1, 

    move(M,X,Z,Y), 

    move(1,X,Y,_), 

    move(M,Z,Y,X).  





% Problema delle 4 regine

% stato: [x1,x2,x3,x4] (posizione delle regine in ogni riga)



solve(P) :-

     perm([1,2,3,4,5,6,7,8],P), 

     combine([1,2,3,4,5,6,7,8],P,S,D),

     all_diff(S),

     all_diff(D).



combine([X1|X],[Y1|Y],[S1|S],[D1|D]) :-

     S1 is X1 +Y1,

     D1 is X1 - Y1,

     combine(X,Y,S,D).

combine([],[],[],[]).



all_diff([X|Y]) :-  \+member(X,Y), all_diff(Y).

all_diff([X]).



perm([X|Y],Z) :- perm(Y,W), takeout(X,Z,W). perm([],[]). 

















% Best-first search assuming strong admissibility and the search is on a tree.



best_first_search_st_ad_tree(Start, Hfun, GoalPred, Sol) :-



bstfs_st_ad_tree([0-[Start]], [ ], Hfun, GoalPred, Sol).



bstfs_st_ad_tree([ _ -[Node | Path] | _ ], _ , _ , GoalPred, [Node | Path]) :-

Goal =.. [GoalPred,Node], Goal. 



bstfs_st_ad_tree([Val-[Node | Path] | MoreOPEN], CLOSED, Hfun, GoalPred, Sol) :-

findall( Val1-[Next,Node | Path],

(tree_arc(Node, Next),

P =.. [Hfun, [Next,Node | Path], Val1], P ),

Neighbors ), % note difference: append(Neighbors, MoreOPEN, TmpOPEN1),

move_smallest_to_top(TmpOPEN1, NewOPEN),

bstfs_st_ad_tree(NewOPEN, [Val-Node | CLOSED], Hfun, GoalPred, Sol). 



�/******************************************************

  CLAUSES FOR DIFFERENTIATION

******************************************************/

d(int(_),_,int(0)).

d(var(X),X,int(1)):-!.

d(var(_),_,int(0)).

d(plus(U,V),X,plus(U1,V1)):-

        d(U,X,U1),

        d(V,X,V1).

d(minus(U,V),X,minus(U1,V1)):-

        d(U,X,U1),

        d(V,X,V1).

d(mult(U,V),X,plus(mult(U1,V),mult(U,V1))):-

        d(U,X,U1),

        d(V,X,V1).

d(div(U,V),X,div(minus(mult(U1,V),mult(U,V1)),mult(V,V))):-

        d(U,X,U1),

        d(V,X,V1).

d(ln(U),X,mult(div(int(1),U),U1)):-d(U,X,U1).

d(potens(E1,int(I)),X,mult(mult(int(I),potens(E1,int(I1))),EXP)):- 

        I1=I-1, 

        d(E1,X,EXP). 

d(sin(U),X,mult(cos(U),U1)):-d(U,X,U1). 

d(cos(U),X,minus(int(0),mult(sin(U),U1))):-d(U,X,U1).

d(tan(U),X,mult(potens(sec(U),int(2)),U1)):-d(U,X,U1).

  



/******************************************************

  CLAUSES FOR SCANNING THE STRING

******************************************************/

tokl(STR,[TOK|TOKL]):-

        fronttoken(STR,TOK,STR1),!,

        tokl(STR1,TOKL).

tokl(_,[]).





/******************************************************

  CLAUSES FOR PARSING OF AN EXPRESSION

******************************************************/

s_exp(IL,OL,EXP):-plusexp(IL,OL,EXP).



plusexp(IL,OL,EXP2):-

        multexp(IL,OL1,EXP1),

        plusexp1(OL1,OL,EXP1,EXP2).



plusexp1(["+"|IL],OL,EXP1,EXP3):-!,

        multexp(IL,OL1,EXP2),

        plusexp1(OL1,OL,plus(EXP1,EXP2),EXP3).

plusexp1(["-"|IL],OL,EXP1,EXP3):-!,

        multexp(IL,OL1,EXP2),

        plusexp1(OL1,OL,minus(EXP1,EXP2),EXP3).

plusexp1(IL,IL,EXP,EXP).



multexp(IL,OL,EXP2):-

        potensexp(IL,OL1,EXP1),

        multexp1(OL1,OL,EXP1,EXP2).



multexp1(["*"|IL],OL,EXP1,EXP3):-!,

        potensexp(IL,OL1,EXP2),

        multexp1(OL1,OL,mult(EXP1,EXP2),EXP3).

multexp1(["/"|IL],OL,EXP1,EXP3):-!,

        potensexp(IL,OL1,EXP2),

        multexp1(OL1,OL,div(EXP1,EXP2),EXP3).

multexp1(IL,IL,EXP,EXP).



potensexp(IL,OL,EXP2):-

        elmexp(IL,OL1,EXP1),

        potensexp1(OL1,OL,EXP1,EXP2),!.

potensexp1(["^"|IL],OL,EXP1,EXP3):-!,

        elmexp(IL,OL1,EXP2),

        potensexp1(OL1,OL,potens(EXP1,EXP2),EXP3).

potensexp1(IL,IL,EXP,EXP).



elmexp(["("|IL],OL,EXP):-

        s_exp(IL,OL1,EXP),

        front(")",OL1,OL),!.

elmexp(["ln","("|IL],OL,ln(EXP)):-

        s_exp(IL,OL1,EXP),

        front(")",OL1,OL),!.

elmexp(["sin","("|IL],OL,sin(EXP)):-

        s_exp(IL,OL1,EXP),

        front(")",OL1,OL),!.

elmexp(["cos","("|IL],OL,cos(EXP)):-

        s_exp(IL,OL1,EXP),

        front(")",OL1,OL),!.

elmexp(["tan","("|IL],OL,tan(EXP)):-

        s_exp(IL,OL1,EXP),

        front(")",OL1,OL),!.

elmexp(["-",TALSTR|IL],IL,int(INT)):-!,

        str_int(TALSTR,INTp),

        INT = -INTp.

elmexp([TALSTR|IL],IL,int(INT)):-str_int(TALSTR,INT),!.

elmexp([NAME|IL],IL,var(NAME)).



front(TOK,[TOK|L],L).





/******************************************************

  CLAUSES FOR REDUCTION OF AN EXPRESSION

******************************************************/

reduce(plus(X,Y),R):-!,

        reduce(X,X1),

        reduce(Y,Y1),

        plusr(X1,Y1,R).

reduce(minus(X,Y),R):-!,

        reduce(X,X1),

        reduce(Y,Y1),

        minusr(X1,Y1,R).

reduce(mult(X,Y),R):-!,

        reduce(X,X1),

        reduce(Y,Y1),

        multr(X1,Y1,R).

reduce(div(X,Y),R):-!,

        reduce(X,X1),

        reduce(Y,Y1),

        divr(X1,Y1,R).

reduce(ln(X),R):-!,

        reduce(X,X1),

        lnr(X1,R).

reduce(potens(E,int(1)),E):-!.

reduce(R,R).



%  CLAUSES FOR REDUCTION OF AN ADDITION EXPRESSION

plusr(int(0),X,X):-!.

plusr(X,int(0),X):-!.

plusr(int(X),int(Y),int(Z)):-!,

        X+Y=Z.

plusr(X,X,mult(int(2),X)):-!.

plusr(int(X),Y,Z):- 

        X<0,

        T=-X,!,

        minusr(int(T),Y,Z).

plusr(Y,int(X),Z):-  

        X<0,

        T=-X,!,

        minusr(int(T),Y,Z).

plusr(mult(int(I),X),X,mult(int(I1),X)):-!,

        I+1=I1.

plusr(X,mult(int(I),X),mult(int(I1),X)):-!,

        I+1=I1.

plusr(mult(int(I1),X),mult(int(I2),X),mult(int(I3),X)):-!,

        I1+I2=I3.

plusr(int(I),X,plus(X,int(I))):-!.

plusr(plus(X,int(I1)),int(I2),plus(X,int(I3))):-!,

        I1+I2=I3.

plusr(plus(X,int(I1)),plus(Y,int(I2)),plus(R,int(I3))):-!,

        I1+I2=I3,

        plusr(X,Y,R).

plusr(plus(X,int(I)),Y,plus(R,int(I))):-!,

        plusr(X,Y,R).

plusr(X,Y,plus(X,Y)).



%  CLAUSES FOR REDUCTION OF A MINUS EXPRESSION

minusr(int(X),int(Y),int(Z)):-!,Z=X-Y.

minusr(X,int(0),X):-!.

minusr(X,X,int(0)):-!.

minusr(X,int(I),plus(int(I1),X)):-!,I1=-I.

minusr(X,Y,minus(X,Y)).



% CLAUSES FOR REDUCTION OF A MULTIPLICATION EXPRESSION

multr(int(X),int(Y),int(Z)):-!,X*Y=Z.

multr(int(0),_,int(0)):-!.

multr(_,int(0),int(0)):-!.

multr(int(1),X,X):-!.

multr(X,int(1),X):-!.

multr(M,plus(X,Y),plus(X1,Y1)):-!,

        multr(M,X,X1),multr(M,Y,Y1). 

multr(M,minus(X,Y),minus(X1,Y1)):-!,

        multr(M,X,X1),multr(M,Y,Y1). 

multr(plus(X,Y),M,plus(X1,Y1)):-!,

        multr(M,X,X1),multr(M,Y,Y1). 

multr(minus(X,Y),M,minus(X1,Y1)):-!,

        multr(M,X,X1),multr(M,Y,Y1). 

multr(mult(int(I1),X),int(I2),M1):-!,

        I1*I2=I3,

        multr(int(I3),X,M1).

multr(int(I1),mult(int(I2),X),M1):-!,

        I1*I2=I3,

        multr(int(I3),X,M1).

multr(mult(int(I1),X),mult(int(I2),Y),mult(int(I3),R)):-!,

        I1*I2=I3, 

        multr(X,Y,R).

multr(mult(int(I),X),Y,mult(int(I),R)):-!,

        multr(X,Y,R).

multr(X,int(I),mult(int(I),X)):-!.

multr(potens(X,int(I1)),potens(X,int(I2)),potens(X,int(I3))):-!,

        I3=I1+I2.

multr(X,potens(X,int(I)),potens(X,int(I1))):-!,

        I1=I+1.

multr(potens(X,int(I)),X,potens(X,int(I1))):-!,

        I1=I+1.

multr(X,X,potens(X,int(2))):-!.

multr(X,Y,mult(X,Y)).



%  CLAUSES FOR REDUCTION OF A DIVISION EXPRESION

divr(int(0),_,int(0)):-!.

divr(_,int(0),var("'endless'")):-!,write("division by zero"),nl.

divr(X,int(1),X):-!.

divr(X,Y,div(X,Y)).



%  CLAUSES FOR REDUCTION OF A LOGARITHM EXPRESSION

lnr(int(0),var("endless")):-!,write("logarithm error"),nl.

lnr(int(1),int(0)):-!.

lnr(X,ln(X)).







/******************************************************

  CLAUSE FOR WRITING OF AN EXPRESSION

******************************************************/

writeexp(EXP):-

        strexp(EXP,STR),

        write(STR).



/*

  CLAUSES FOR CONVERTING AN EXPRESSION TO A STRING

*/

% Taken from the old writeexp clauses



  strexp(var(NAME),NAME).

  strexp(int(INT),INTSTR) :-

      str_int(INTSTR,INT).

  strexp(ln(EXP),STR)  :-

      strPAR(EXP,STRp),

      concat("ln",STRp,STR).

  strexp(sin(EXP),STR) :- 

      strPAR(EXP,STRp),

      concat("sin",STRp,STR).

  strexp(cos(EXP),STR) :- 

      strPAR(EXP,STRp),

      concat("cos",STRp,STR).

  strexp(tan(EXP),STR) :- 

      strPAR(EXP,STRp),

      concat("tan",STRp,STR).

  strexp(sec(EXP),STR) :- 

      strPAR(EXP,STRp),

      concat("sec",STRp,STR).

  strexp(plus(EXP1,EXP2),STR):-

      strexp(EXP1,STR1),

      concat(STR1,"+",STR3),

      strexp(EXP2,STR2),

      concat(STR3,STR2,STR).

  strexp(minus(EXP1,EXP2),STR):-

      strexp(EXP1,STR1),

      concat(STR1,"-",STR3),

      strMINUS(EXP2,STR2),

      concat(STR3,STR2,STR).

  strexp(mult(EXP1,EXP2),STR):-

      strMINUS(EXP1,STR1),

      concat(STR1,"*",STR3),

      strMULT(EXP2,STR2),

      concat(STR3,STR2,STR).

  strexp(div(EXP1,EXP2),STR):-

      strMULT(EXP1,STR1),

      concat(STR1,"/",STR3),

      strDIV(EXP2,STR2),

      concat(STR3,STR2,STR).

  strexp(potens(EXP1,EXP2),STR):-

      strDIV(EXP1,STR1), 

      concat(STR1,"^",STR3),

      strPOTENS(EXP2,STR2),

      concat(STR3,STR2,STR).



  strPOTENS(div(X,Y),STR):-!,strPAR(div(X,Y),STR).

  strPOTENS(X,STR):-strDIV(X,STR).



  strDIV(mult(X,Y),STR):-!,strPAR(mult(X,Y),STR).

  strDIV(X,STR):-strMULT(X,STR).



  strMULT(minus(X,Y),STR):- !,strPAR(minus(X,Y),STR).

  strMULT(X,STR):-strMINUS(X,STR).



  strMINUS(plus(X,Y),STR):-!,strPAR(plus(X,Y),STR).

  strMINUS(X,STR):-strexp(X,STR).



  strPAR(EXP,STR):-

      strexp(EXP,STR1),

      concat("(",STR1,STR2),

      concat(STR2,")",STR).

  

�/* From the book PROLOG PROGRAMMING IN DEPTH

   by Michael A. Covington, Donald Nute, and Andre Vellino.

   Copyright 1988 Scott, Foresman & Co.

/* Modified for Quintus Prolog by Andreas Siebert */



/* CANNIBAL.PL */



/*

 * Main predicate

 */



missionaries_and_cannibals :-

     ferry([[3,3,l,0,0]],Solution),

     fast_reverse(Solution,ReversedSolution),

     show_ferry(ReversedSolution).



/*

 * How to move people across the river

 */



ferry([OldSit|Rest],Solution) :-

     safe_trip(OldSit,NewSit),

     \+ member(NewSit,Rest),

     check_ferry([NewSit,OldSit|Rest],Solution).



safe_trip([ML,CL,l,MR,CR],[MLL,CLL,r,MRR,CRR]) :-

     in_boat(M,C,ML,CL),

     MLL is ML - M,

     CLL is CL - C,

     not_overpowered(MLL,CLL),

     MRR is MR + M,

     CRR is CR + C,

     not_overpowered(MRR,CRR).



safe_trip([ML,CL,r,MR,CR],[MLL,CLL,l,MRR,CRR]) :-

     in_boat(M,C,MR,CR),

     MLL is ML + M,

     CLL is CL + C,

     not_overpowered(MLL,CLL),

     MRR is MR - M,

     CRR is CR - C,

     not_overpowered(MRR,CRR).



/*

 * How to find out whether we've finished

 */



check_ferry([[0,0,r,3,3]|Rest],[[0,0,r,3,3]|Rest]).



check_ferry(SequenceOfSituations,Solution) :-

     ferry(SequenceOfSituations,Solution).



/*

 * Other constraints on putting people in the boat

 */



in_boat(M,C,MM,CC) :- boat_load(M,C), M =< MM, C =< CC.



boat_load(1,0).

boat_load(0,1).

boat_load(1,1).

boat_load(2,0).

boat_load(0,2).



not_overpowered(0,_).

not_overpowered(M,C) :- M >= C.



/*

 * List processing utilities

 */



fast_reverse(X,Y) :- fast_reverse_aux(X,Y,[]).



fast_reverse_aux([],X,X).

fast_reverse_aux([H|T],Result,Temp) :-

     fast_reverse_aux(T,Result,[H|Temp]).



member(X,[X|_]).

member(X,[_|Y]) :- member(X,Y).



/*

 * Routines to display the solution as a picture

 */



show_ferry([]).



show_ferry([[ML,CL,B,MR,CR]|Rest]) :-

     nl, nl,

     write_times('M ',ML),

     write_times('C ',CL),

     boat(B,Picture),

     write(Picture),

     write_times('M ',MR),

     write_times('C ',CR),

     show_ferry(Rest).



write_times(_,0).



write_times(X,N) :-      /* write X, N times */

     write(X),

     M is N - 1,

     write_times(X,M).



boat(l,'\(___)                    / ').

boat(r,'\                    (___)/ ').

